

A mechanized model of Type Theory based on an internalization of groupoids

Matthieu Sozeau^{1,2} and Nicolas Tabareau^{1,3}

¹ πr^2 and Ascola teams, INRIA

² Preuves, Programmes et Systèmes (PPS)

³ Laboratoire d'Informatique de Nantes Atlantique (LINA)

`firstname.surname@inria.fr`

Abstract. Homotopical interpretations of Martin-Löf type theory lead toward an interpretation of equality as a richer, more extensional notion. Extensional or axiomatic presentations of the theory with principles based on such models do not yet fully benefit from the power of dependent type theory, that is its computational character. Reconciling intensional type theory with this richer notion of equality requires to move to higher-dimensional structures where equality reasoning is explicit. In this paper, we follow this idea and develop an internalization of a model of Martin-Löf type theory with functional extensionality based on groupoids and respecting the invariance by isomorphism principle. Our formal development relies crucially on ad-hoc polymorphism to overload notions of equality and on a conservative extension of Coq's universe mechanism with polymorphism.

1 Introduction

A notorious difficulty with intensional type theories like Martin-Löf type theory or the calculus of inductive constructions is the lack of extensionality principles in the core theory, and notably in its notion of propositional equality. This makes the theory both inflexible with respect to the notion of equality one might want to use on a given type and also departs from the traditional equality notion from set theory. Functional extensionality (pointwise equal functions are equal), propositional extensionality (logically equivalent propositions are equal) and proof irrelevance (all proofs of the same proposition are equal) are principles that are valid extensions of type theory and can be internalized in an intensional theory, in the manner of Altenkirch et al. [1]. Another extensionality principle coming from homotopy theory is univalence, whose first instance is the idea that isomorphic types are equal [2,3]. All these principles should be imposable on a type theory like MLTT because, intuitively, no term can distinguish between isomorphic types, pointwise equal functions, logically equivalent propositions or two proofs of the same proposition. This hints at the idea that there ought to exist an internal model of type theory where equality is defined on a type by type basis using the aforementioned principles and a translation that witnesses

that any term of the theory is equipped with proofs that they respect these properties. Formalizing this definitional translation is the goal of this paper.

The central change in the theory is in the treatment of equality. Much interest has been devoted to the study of the identity type of type theory and models thereof, starting with the groupoid model of Hofmann and Streicher [4]. This eventually led to the introduction of homotopy type theory and the study of the ω -groupoid model of type theory with identity types, which validates extensionality principles. This model in turn guides work to redesign type theory itself to profit from its richness, and develop a new theory that internalizes the new principles. Preliminary attempts have been made, notably by Licata and Harper [5] who develop a 2-dimensional version of an hybrid intensional/extensional type theory which integrates functional extensionality and univalence in the definition of equality. Work is underway to make it more intensional, and this starts by making the step to higher dimensions, whether finite (weak n -groupoids) or infinite (weak ω -groupoids) [6]. Our work here concentrates on the internalization in COQ of Hofmann and Streicher’s groupoid model where we can have a self-contained definition of the structures involved. Our first motivation to implement this translation is to explore the interpretation of type theory in groupoids in a completely intensional setting and in the type theoretic language, leaving no space for imprecision on the notions of equality and coherence involved. We also hope to give with this translation a basic exposition of the possible type theoretic implications of the groupoid/homotopy models, bridging a gap in the literature. On the technical side, we nevertheless have to slightly move away from Hofmann and Streicher interpretation. In particular, it is not possible to interpret a univalent universe in the internalization of their model; and this for two important reasons that we now discuss.

A univalent model that makes no use of univalence. Our long term goal is to provide an interpretation of homotopy type theory into type theory (without extensional principle). This would allow to give a meaning to all extensionality principles without relying on them in the target theory.

However, if we use a traditional approach and formalize groupoid laws using the identity type, it turns out that the type of isomorphisms between two objects x and y of a groupoid, noted $x \sim_1 y$, must also be formalized using the identity type. Or, putting it in the homotopy type theoretic language, we have to consider univalent groupoids. This is because, in the groupoid interpretation, we need isomorphisms between groupoids to satisfy groupoids laws using the identity type as notion of equality between isomorphisms. Then, using functional extensionality, this amounts to compare the two functions pointwise, but again using the identity type, and not the internal notion of equality in the groupoid, given by \sim_1 . This means that \sim_1 has to coincide with the identity type, which is precisely the property of being a univalent groupoid. But then, this means that isomorphisms between groupoids should be reflected in the identity type, which forces the target theory to satisfy univalence already. . . So the first conclusion is that an internalization of the groupoid interpretation in the style of formaliza-

tion of categories presented in [7] would require the target theory to be univalent already.

To avoid this issue in our internalization of groupoids, the groupoid laws are imposed using a notion of 2-dimensional equality that does not have to be the identity type, but just to be an equivalence relation. Then, to enforce that the types of isomorphisms constitute (homotopical) sets, we still use the identity type, but only to express triviality of higher dimensions, not coherences themselves. This interpretation of strictness is closer to the idea that a groupoid is a weak ω -groupoid for which all equalities at dimension 2 are the same. Note that our presentation requires less properties on identity types, but we still need the axiom of functional extensionality to prove triviality of higher dimensions for the groupoid of functors. Also, this indicates that if we scale to ω -groupoids, the presence of identity types in the core type theory will not be necessary anymore and so the core type theory will be axiom free. Thus, this paper can be seen as a proof of concept that it is possible to interpret homotopy type theory into type theory without identity types.

Taking size issue into account. In Hofmann and Streicher groupoid model, a type A depending on context Γ is interpreted as a functor

$$\llbracket A \rrbracket : \llbracket \Gamma \rrbracket \longrightarrow \mathbf{Gpd}$$

where \mathbf{Gpd} is said to be “*the groupoid of groupoids*”. But groupoids form actually a 2-groupoids and not a groupoid, because the type of isomorphisms is not a set. That is groupoids are not enriched over themselves. Hofmann and Streicher have solves this issue as we usually do in category theory, by considering small groupoids, that is groupoids for which the type of objects is also a set. This is fine as long as we consider set theory. But when moving to homotopy type theory, small groupoids actually correspond to setoids. [NT] ►*expand on this ?*◄ This means that, using an internalization of groupoids, one can only interpret types as setoids, and not as groupoids. This is what we do in this paper, which prevent use to give a complete formulation of univalence in the source theory. That is, the invariance by isomorphism principle, also respected in the model, as no counterpart in the source theory.

Another way to solve this size issue would be to formalize the notion of 2-groupoids instead but we do not think that the interpretation would have gained much. We believe that the real challenge is to interpret ω -groupoids which is the proper notion of self-enriched groupoid-like structure and is the subject of ongoing works.

Outline of the paper. Section 2 introduces the source type theory of the translation and some features of the proof assistant that are used in the formalization. The formal model includes a formalization of groupoids and associated structures (§3.2-3.5) and a construction of the groupoids interpreting the standard type constructors (§3.6-3.7). Section 4 presents a semi-formal interpretation of the dependent type theory in the groupoid model and gives proofs that our interpretation validates a univalence principle at the level of sets (§4.4).

2 Setting of the translation

2.1 Martin-Löf Type Theory with Functional Extensionality

For the purpose of this paper we study a restricted source theory resembling a cut-down version of the core language of the COQ system, with only one **Type** universe (see [8] for an in-depth study of this system). This is basically Martin-Löf Type Theory (without **Type : Type**). First we introduce the definitions of the various objects at hand. We start with the term language of a dependent λ -calculus: we have a countable set of variables x, y, z , and the usual typed λ -abstraction $\lambda x : \tau, b$, à la Church, application $t u$, the dependent product and sum types $\Pi/\Sigma x : A. B$, and an identity type $\text{Id}_T t u$.

The typing judgment for this calculus is written $\Gamma \vdash t : T$ (Figure 1) where Γ is a context of declarations $\Gamma ::= \cdot \mid \Gamma, x : \tau, t$ and T are terms. If we have a valid derivation of this judgment, we say t has type T in Γ .

Most of the rules are standard. The definitional equality $A = B$ is defined as the congruence on type and term formers compatible with β -reductions for abstractions and projections.

Identity type. The identity type in the source theory is the standard Martin-Löf identity type $\text{Id}_T t u$, generated from an introduction rule for reflexivity with the usual J eliminator and its *propositional* reduction rule. The J reduction rule will actually be valid definitionally in the model for *closed* terms.

Functional Extensionality. The proof-relevant equality of functions in the source theory for which we want to give a computational model appears in rule FUN-EXT, extending the formation rules of identity types on dependent product. Equality of dependent functions f and g of type $\Pi x : A. B$ are introduced by giving a witness of $\Pi t : A, \text{Id}_{B\{t/x\}} (f t) (g t)$. The J rule for dependent functions witnesses the *naturality* of every predicate constructed in the source type theory. This rule corresponds to the introduction of equality on dependent functions in [5].

2.2 The proof assistant

We use an extension of the COQ proof assistant to formally define our translation. Vanilla features of COQ allow us to define overloaded notations and hierarchies of structures through type classes [9], and to separate definitions and proofs using the PROGRAM extension [10], they are both documented in COQ's reference manual [11]. We also use the recent extension to polymorphic universes [12].

Classes and projections. The formalization makes heavy use of type classes and sigma types, both defined internally as parameterized records. We also have modified the representation of record projections, making them primitive to allow a more economical representation, leaving out the parameters of the record type they are applied to. This change, which is justified by bidirectional presentations of type theory, makes typechecking exponentially faster in the case of nested structures (see [13] for a detailed explanation of this phenomenon).

$$\begin{array}{c}
\text{EMPTY} \quad \frac{}{\cdot \vdash} \quad \text{DECL} \quad \frac{\Gamma \vdash T \text{ type} \quad x \notin \Gamma}{\Gamma, x : T \vdash} \quad \text{UNIV} \quad \frac{}{\Gamma \vdash \mathcal{U} \text{ type}} \quad \text{VAR} \quad \frac{\Gamma \vdash (x : T) \in \Gamma}{\Gamma \vdash x : T} \\
\\
\text{PROD/SIGMA} \quad \frac{\Gamma, x : A \vdash B \text{ type}}{\Gamma \vdash \Pi / \Sigma x : A. B \text{ type}} \quad \text{PAIR} \quad \frac{\Gamma \vdash t : A \quad \Gamma \vdash u : B\{t/x\}}{\Gamma \vdash (t, u)_{x:A.B} : \Sigma x : A. B} \quad \text{PROJ1} \quad \frac{\Gamma \vdash t : \Sigma x : A. B}{\Gamma \vdash \pi_1 t : A} \\
\\
\text{PROJ2} \quad \frac{\Gamma \vdash t : \Sigma x : A. B}{\Gamma \vdash \pi_2 t : B\{\pi_1 t/x\}} \quad \text{CONV} \quad \frac{\Gamma \vdash t : A \quad \Gamma \vdash B \text{ type} \quad A = B}{\Gamma \vdash t : B} \\
\\
\text{LAM} \quad \frac{\Gamma, x : A \vdash t : B}{\Gamma \vdash \lambda x : A. t : \Pi x : A. B} \quad \text{APP} \quad \frac{\Gamma \vdash t : \Pi x : A. B \quad \Gamma \vdash t' : A}{\Gamma \vdash t t' : B\{t'/x\}} \\
\\
\text{ID} \quad \frac{\Gamma \vdash T \text{ type} \quad \Gamma \vdash A, B : T}{\Gamma \vdash \text{Id}_T A B \text{ type}} \quad \text{ID-INTRO} \quad \frac{\Gamma \vdash t : T}{\Gamma \vdash \text{refl}_T t : \text{Id}_T t t} \quad \text{FUN-EXT} \quad \frac{\Gamma \vdash e : \Pi t : A, \text{Id}_{B\{t/x\}} (f t) (g t)}{\Gamma \vdash \text{fun_ext}(e) : \text{Id}_{\Pi x:A.B} f g} \\
\\
\text{ID-ELIM (J)} \quad \frac{\Gamma \vdash i : \text{Id}_T t u \quad \Gamma, x : T, e : \text{Id}_T t x \vdash P \text{ type} \quad \Gamma \vdash p : P\{t/x\}\{\text{refl}_T t/e\}}{\Gamma \vdash \text{J}_{\lambda x. e. P} i p : P\{u/x, i/e\}}
\end{array}$$

Fig. 1: Typing judgments for our extended MLTT

One peculiarity of COQ's class system we use is the ability to nest classes. We use the $\text{A_of_B} := A$ notation in a type class definition Class B as an abbreviation for defining A_of_B as an instance of A .

Polymorphic Universes. To typecheck our formalization, we also need a stronger universe system than what vanilla COQ offers. Indeed, if we are to give a uniform (shallow) translation of type theory in type theory, we have to define a translation of the type universe (a groupoid) as a term of the calculus and equip type constructors like Π and Σ with $\llbracket \text{type} \rrbracket$ structures as well. As $\llbracket \text{type} \rrbracket$ itself contains a Type , the following situation occurs when we define the translation of, e.g. sums: we should have $\llbracket \Sigma U T \text{ type} \rrbracket = \llbracket \Sigma \rrbracket \llbracket U \rrbracket \llbracket T \rrbracket : \llbracket \text{type} \rrbracket$. To ensure consistency of the interpretations of types inside $\llbracket U \rrbracket$, $\llbracket T \rrbracket$ and the outer one, they must be at different levels, with the outer one at least as large as the inner ones. A universe polymorphic extension of COQ has been designed to allow such highly generic developments [12]. The design was implemented by the first author and is already in use to check the HoTT library in COQ [14].

3 Formalization of groupoids

This section presents our formalization of groupoids in Coq with universe polymorphism. We first explain our overloaded management of equalities and introduce type classes for groupoids and their associated structures, i.e., functors, natural transformations and homotopy equivalences (§3.2-3.4). Natural transformations give access to a homotopic form of functional extensionality, while homotopy equivalences provide extensionality at the level of 0-types. Polymorphic universes are needed to state that setoids and homotopy equivalences form a groupoid. Homotopic equivalences directly provide access to a rewriting mechanism on types (§3.5). This rewriting is used to extend functors and products to dependent functors and dependent sums (§3.6-3.7).

3.1 Notations

We use the following notations throughout: Sigma type introduction is written ($t ; p$) when its predicate/fibration is inferrable from the context, and projections are denoted π_1 and π_2 . The bracket notation $[-]$ is an alias for π_1 . The following is directly extracted from Coq .v files using the `coqdoc` tool (source files are available at <https://github.com/mattam82/groupoid>). If you are reading the colored version of the paper, keywords are typeset in red, inductive types and classes in blue, inductive constructors in dark red, and defined constants and lemmas in green.

3.2 Definition of groupoids

We formalize groupoids using type classes. Contrarily to what is done in the setoid translation, the basic notion of a morphism is an inhabitant of a relation in **Type** (i.e., a proof-relevant relation):

Definition `HomSet` ($T : \text{Type}$) := $T \rightarrow T \rightarrow \text{Type}$.

Given x and y in T , `HomSet` T x y is the type of morphism from x to y . To manipulate different `HomSet`'s at dimension 1 and 2 more abstractly, we use ad-hoc polymorphism and introduce type classes `HomSet1` and `HomSet2` with according notations.

Class `HomSet1` T := {`eq1` : `HomSet` T }.

Infix `"~1"` := (`.(eq1)`) (at level 80).

Class `HomSet2` { T } (`Hom` : `HomSet` T) := {`eq2` : $\forall \{x\ y : T\}, \text{HomSet } (\text{Hom } x\ y)$ }.

Infix `"~2"` := `eq2` (at level 80).

Given a `HomSet`, we define type classes: `Identity` that gives the identity morphism, `Inverse` which corresponds to the existence of an inverse morphism for every morphism (noted f^{-1}) and `Composition` which corresponds to morphism composition (noted $g \circ f$). Those three properties are gathered by the type class `Equivalence`. A `PreCategory` is defined as a category where coherences are given up-to an equivalence relation denoted by `~2`. Ordinary categories are derived with the additional requirement that higher equalities are trivial, which can be expressed using identity types (see the definition of `lsType1`).

We do not put this condition into the basic definition because categories and functors form a pre-category but not a 1-category. Thus, working with pre-categories and pre-groupoids allows to share more structure and is closer to the ω -groupoid model which is itself enriched.

```

Class PreCategory T := { Hom1 :> HomSet1 T; Hom2 :> HomSet2 eq1;
  Id :> Identity eq1; Comp :> Composition eq1;
  Equivalence2 :> ∀ x y, (Equivalence (eq2 (x:=x) (y:=y))));
  idR : ∀ x y (f : x ~1 y), f ∘ identity x ~2 f ;
  idL : ∀ x y (f : x ~1 y), identity y ∘ f ~2 f ;
  assoc : ∀ x y z w (f : x ~1 y) (g : y ~1 z) (h : z ~1 w),
    (h ∘ g) ∘ f ~2 h ∘ (g ∘ f);
  comp : ∀ x y z (f f' : x ~1 y) (g g' : y ~1 z),
    f ~2 f' → g ~2 g' → g ∘ f ~2 g' ∘ f' }.
```

In homotopy type theory, coherences are expressed using identity types, with a further requirement that the internal notion of equality induced by the category (isomorphism between two objects) coincides with its identity type. We do not share this point of view because our goal is to restrict the use of identity types to the treatment of contractedness for higher cells. Note that the `comp` law is not present in traditional definition of categories because it is automatically satisfied for the identity type.

A `PreGroupoid` is a `PreCategory` where all 1-Homs are invertible and subject to additional compatibility laws for inverses.

```

Class PreGroupoid T := { C :> PreCategory T ; Inv :> Inverse eq1 ;
  invR : ∀ x y (f : x ~1 y), f ∘ f-1 ~2 identity y ;
  invL : ∀ x y (f : x ~1 y), f-1 ∘ f ~2 identity x ;
  inv : ∀ x y (f f' : x ~1 y), f ~2 f' → f-1 ~2 f'-1 }.
```

Groupoids are then pre-groupoids where equality at dimension 2 is irrelevant. This irrelevance is defined using a notion of contractibility expressed with (relevant) identity types.

This is a way to require that all higher-cells are trivial. In our setting, we do not have the possibility to say that all 2-cells are related by a 3-cell, and so on. The price to pay will be explicit reasoning on identity types when proving for instance contractedness for the function space. In that case, we need the axiom of functional extensionality. By analogy to homotopy type theory, we note `IsType1` the property of being a groupoid.

```

Class IsType1 T := { G :> PreGroupoid T ;
  is_Trunc2 : ∀ (x y : T) (e e' : x ~1 y) (E E' : e ~2 e'), Contr (E = E') }.
```

In the same way, we define `IsType0` when equality is irrelevant at dimension 1.

```

Class IsType0 T := { S :> IsType1 T ;
  is_Trunc1 : ∀ (x y : T) (e e' : x ~1 y), Contr (e = e') }.
```

We note `Type1` for the types that form a `IsType1`. The subscript 1 comes from the fact that groupoids are 1-truncated types in homotopy type theory. In the same way, we note `Type0` for the types that form a `IsType0`. We define $T|_s$ the lifting of setoids to groupoids.

3.3 Functors and natural transformations

A morphism between two groupoids is a functor, i.e., a function between objects of the groupoids that transports homs and subject to compatibility laws.

```
Class Functor { T U : Type1 } (f : [T] → [U]) : Type :=
{ map : ∀ {x y}, x ~1 y → f x ~1 f y ;
  mapid : ∀ {x}, map (identity x) ~2 identity (f x) ;
  mapcomp : ∀ {x y z} (e : x ~1 y) (e' : y ~1 z), map (e' ∘ e) ~2 map e' ∘ map e ;
  map2 : ∀ {x y : [T]} {e e' : x ~1 y}, (e ~2 e') → map e ~2 map e' }.
```

Definition `Fun_Type` (T U : Type₁) := {f : [T] → [U] & Functor f}.

We note $T \rightarrow U$ the type of functors from T to U . Note that we only impose compatibility with the composition as compatibilities with identities and inverse Homs can be deduced from it. We note $M \star N$ the application of a function M in the first component of a dependent pair. Equivalence between functors is given by natural transformations. We insist here that this naturality condition in the definition of functional equality is crucial in a higher setting. It is usually omitted in formalizations of homotopy theory in Coq because there they only consider the 1-groupoid case where the naturality becomes trivial, see for instance [15].

```
Class NaturalTrans T U {f g : T → U} (α : ∀ t : [T], f ★ t ~1 g ★ t) :=
  αmap : ∀ {t t'} (e : t ~1 t'), α t' ∘ map f e ~2 map g e ∘ α t.
```

Definition `nat_trans` T U : HomSet (T → U)
:= λ f g, {α : ∀ t : [T], f ★ t ~₁ g ★ t & NaturalTrans α}.

In our setting, equality between natural transformations is not expressed using identity types, but using the higher categorical notion of modification.

Definition `modification` T U (f g : T → U) : HomSet (f ~₁ g)
:= λ α β, ∀ t : [T], α ★ t ~₂ β ★ t.

We can now equip the functor space with a groupoid structure. Note here that we (abusively) use the same notation for the functor type and its corresponding groupoid.

Definition `_fun` T U : Type₁ := (T → U ; fun_{grp} T U).

In the definition above, `fungrp` is a proof that `nat_trans` and `modification` form a groupoid on $T \rightarrow U$. In particular, it makes use of functional extensionality, which says that the canonical proof of $f = g \rightarrow \forall x, f x = g x$ is an equivalence (in the sense of homotopy type theory).

3.4 Homotopic equivalences

The standard notion of equivalence between groupoids is given by adjoint equivalences, that is a map with an `adjoint` and two proofs that they form a `section` (or counit of the adjunction) and a `retraction` (or unit of the adjunction).

```
Class Iso_struct T U (f : [T] → [U]) :=
{ adjoint : [U] → [T] ;
  section : f ∘ adjoint ~2 identity U ;
  retraction : adjoint ∘ f ~2 identity T }.
```

This type class defines usual equivalences. To get an adjoint equivalence, an additional triangle identity between sections and retractions is required. This

allows to eliminate a section against a retraction in proofs. A corresponding triangle identity involving `adjoint f` can also be expressed, but it can be shown that each condition implies the other.

```
Class Equiv_struct T U (f : T → U) :=
{ iso : Iso_struct f;
  triangle : ∀ t, section ★ (f ★ t) ~₂ map f (retraction ★ t)}.
```

```
Definition Equiv A B := {f : A → B & Equiv_struct f}.
```

It is well known that any equivalence can be turned into an adjoint equivalence by slightly modifying the section. While available in our formalization, this result should be used with care as it opacifies the underlying notion of homotopy and can harden proofs.

Equality of homotopy equivalences is given by equivalence of adjunctions. Two adjunctions are equivalent if their left adjoint are equivalent and they agree on their sections (up-to the isomorphism). Note that equivalence of the right adjoints and agreement on their retractions can be deduced so they are not part of the definition.

```
Class EquivEq {T U} {f g : Equiv T U} (α : [f] ~₂ [g]) : Type :=
_eq_section : section f ~₂ section g ∘ (α ∘ (Equiv_adjoint α)).
```

```
Definition Equiv_eq T U (f g : Equiv T U) := {α : nat_trans [f] [g] & EquivEq α}.
```

It is crucial here to be able to express the 2-dimensional equality between groupoids as a particular `Type` and not directly using the identity type. Indeed, whereas the functional extensionality principle makes the use of the identity type and modification equivalent to treat equality of natural transformations, the same is not possible for homotopy equivalences.

We can define the pre-groupoid `Type11` of groupoids and homotopy equivalences. However, groupoids together with homotopy equivalences do not form a groupoid but rather a 2-groupoid. As we only have a formalization of groupoids, this can not be expressed in our setting. Nevertheless, we can state that setoids (inhabitants of `Type0`) form a groupoid.

```
Definition Type01 : Type1 := (Type0 ; Equiv_Type0).
```

In the definition above, `Equiv_Type0` is a proof that `Equiv` and `Equiv_eq` form a groupoid. It makes again use of functional extensionality to prove contractibility of higher cells. As `Type1` appears both in the term and in the type, the use of polymorphic universe is crucial here to avoid an inconsistency.

3.5 Rewriting in homotopy type theory

When considering a dependent family `F` of type `[A → Type11]`, the `map` function provides a homotopy equivalence between `F ★ x` and `F ★ y` for any `x` and `y` such that `x ~1 y`. The underlying map of homotopy equivalence can hence be used to cast any term of type `[F ★ x]` to `[F ★ y]`.

```
Definition transport A (F : [A → Type11]) {x y : [A]} (e : x ~1 y)
: (F ★ x) → (F ★ y) := [map F e].
```

Using compatibility on `map`, we can reason on different transport paths. Intuitively, any two transport maps with the same domain and codomain should be the same up to homotopy. As we only consider groupoids, there is only one

relevant level of compatibilities, higher compatibilities are trivial. $\text{transport}_{\text{eq}}$ is an example of a derivable equality between two transport maps, when the proofs relating x and y are equal.

Definition $\text{transport}_{\text{eq}} A (F:[A \rightarrow \text{Type}_1^1]) \{x\ y:[A]\} \{e\ e':x \sim_1 y\} (H:e \sim_2 e') : \text{transport } F\ e \sim_1 \text{transport } F\ e' := [\text{map}_2\ F\ H]$.

In the text, we also use $\text{transport}_{\text{id}}$, $\text{transport}_{\text{comp}}$ and $\text{transport}_{\text{map}}$ for compatibilities with identities, composition and for the functoriality of transport .

3.6 Dependent Product

As for functions, dependent functions will be interpreted as functors. But this time, the compatibilities with higher-order morphisms cannot be expressed as simple equalities, as some transport has to be done to make those equalities typable. We call such a functor a *dependent functor*. Dependent functors are defined between a groupoid T and a functor U from T to Type_1^1 (the pre-groupoid of groupoids). U must be seen as a type depending on T , or as a family of types indexed by T .

Class $\text{Functor}^T T (U : [T \rightarrow \text{Type}_1^1]) (f : \forall t, [U \star t]) : \text{Type} := \{$
 $\text{map}^T : \forall \{x\ y\} (e : x \sim_1 y), \text{transport } U\ e \star (f\ x) \sim_1 f\ y ;$
 $\text{map}_{\text{id}}^T : \forall x, \text{map}^T (\text{identity } x) \sim_2 \text{transport}_{\text{id}}\ U \star (f\ x);$
 $\text{map}_{\text{comp}}^T : \forall x\ y\ z (e : x \sim_1 y) (e' : y \sim_1 z),$
 $\text{map}^T (e' \circ e) \sim_2 \text{map}^T e' \circ \text{transport}_{\text{map}}\ U _ (\text{map}^T e) \circ$
 $(\text{transport}_{\text{comp}}\ U\ e\ e' \star _);$
 $\text{map}_2^T : \forall x\ y (e\ e' : x \sim_1 y) (H : e \sim_2 e'),$
 $\text{map}^T e \sim_2 \text{map}^T e' \circ (\text{transport}_{\text{eq}}\ U\ H \star (f\ x))\}.$

Definition $\text{II}_T T (U:[T \rightarrow \text{Type}_1^1]) := \{f : \forall t, [U \star t] \ \& \ \text{Functor}^T\ U\ f\}.$

Equality between dependent functors is given by dependent natural transformations and equality at level 2 is given by dependent modifications.

We can now equip dependent functors with a groupoid structure as we have done for functors. We note $\text{II } U$ the dependent product over a family of groupoids U .

A family of setoids can be seen as a family of groupoids using a lifting that we abusively note $U_{\downarrow s}$. We can prove that the dependent product over a family of setoids is also a setoid. We note II_0 the restriction of II to families of setoids.

3.7 Dependent sums

In the interpretation of Σ types, we pay for the fact that we are missing the 2-dimensional nature of Type_1^1 . Indeed, as we will need rewriting in the definition of equality on Σ types, delivering the corresponding groupoid structure requires to reason on compatibility between rewritings, which amount to the missing 2-dimensional laws. However, as Type_0^1 is a groupoid, all 2-dimensional equalities become trivial on a family of setoids, so we can define the groupoid of Σ types over a groupoid T and a morphism of type $[T \rightarrow \text{Type}_0^1]$.

Definition $\Sigma_T T (U : [T \rightarrow \text{Type}_0^1]) := \{t : [T] \ \& \ [U \star t]\}.$

The 1-equality between dependent pairs is given by 1-equality on the first and second projections, with a transport on the second projection on the left.

Definition $\Sigma_{\text{Eq}} T (U : [T \longrightarrow \text{Type}_0^1]) : \text{HomSet} (\Sigma_T U) :=$
 $\lambda m n, \{P : [m] \sim_1 [n] \ \& \ \text{transport} (U_{\uparrow s}) P \star (\pi_2 m) \sim_1 \pi_2 n\}.$

In the same way, 2-equality between 1-equalities is given by projections and rewriting.

Definition $\Sigma_{\text{Eq}_2} T (U : [T \longrightarrow \text{Type}_0^1]) (M N : \Sigma_T U) : \text{HomSet} (M \sim_1 N)$
 $:= \lambda e e', \{P : [e] \sim_2 [e'] \ \& \ \pi_2 e \sim_2 \pi_2 e' \circ (\text{transport}_{\text{eq}} (U_{\uparrow s}) P \star (\pi_2 M))\}.$

This way, we can define the groupoid ΣU of dependent sums for any family of setoids. When T is a setoid, ΣU is also a setoid.

4 The groupoid interpretation

We now organize our formalization of groupoids into a model of the dependent type theory with univalence introduced in Section 2.1. The interpretation is based on the notion of categories with families introduced by Dybjer [16] later used in [4]. This interpretation can also be seen as an extension of the Takeuti-Gandy interpretation of simple type theory, recently generalized to dependent type theory by Coquand et al. using Kan semisimplicial sets or cubical sets [17].

There are two main novelties in our interpretation. First, we take advantage of universe polymorphism to interpret dependent types directly as functors into Type_0^1 . Second, we provide an interpretation in a model where structures that are definitionally equal for Kan semisimplicial sets are only homotopically equal, which requires more care in the definitions (see for instance the definition of Lam in Section 4.3 which mixes two points of view on fibrations).

We only present the computational part of the interpretation, the proofs of functoriality and naturality are not detailed but most of them are available in the COQ development.

4.1 Dependent types

The judgment context $\Gamma \vdash$ of Section 2.1 is represented in COQ as a setoid, noted $\text{Context} := \text{Type}_0$. The empty context (Rule EMPTY) is interpreted as the setoid with exactly one element at each dimension. Types in a context Γ , noted $\text{Typ } \Gamma$, are (context) functors from Γ to the groupoid of setoids Type_0^1 . Thus, a judgment $\Gamma \vdash A : \text{Type}$ is represented as a term A of type $\text{Typ } \Gamma$. Context extension (Rule DECL) is given by dependent sums, i.e., the judgment $\Gamma, x : A \vdash$ is represented as ΣA .

Terms of A introduced by a sequent $\Gamma \vdash x : A$ are dependent (context) functors from Γ to A that return for each context valuation γ , an object of $A \star \gamma$ respecting equality of contexts. The type of terms of A is noted $\text{Tm } A := [\text{II } A]$ (context is implicit).

A dependent type $\Gamma, x : A \vdash B$ is interpreted in two equivalent ways: simply as a type $\text{TypDep } A := \text{Typ} (\Sigma A)$ over the dependent sum of Γ and A or as a type family $\text{TypFam } A$ over A (corresponding to a family of sets in constructive mathematics). A type family can be seen as a fibration (or bundle) from B to A . In what follows, the indice comp is given to proofs of (dependent) functoriality.

Definition $\text{TypFam} \{ \Gamma : \text{Context} \} (A : \text{Typ } \Gamma) :=$

$$[\Pi (\lambda \gamma, (A \star \gamma) \vdash_s \longrightarrow \text{Type}_0^1; \text{TypFam}_{\text{comp}} -)].$$

Terms of $\text{TypDep } A$ and $\text{TypFam } A$ can be related using a dependent closure at the level of types. In the interpretation of typing judgments, this connection will be used to switch between the fibration and the morphism points of view.

Definition $A \{ \Gamma : \text{Context} \} \{ A : \text{Typ } \Gamma \} (B : \text{TypDep } A)$
 $: \text{TypFam } A := (\lambda \gamma, (\lambda t, B \star (\gamma; t) ; -); \Lambda_{\text{comp}} B).$

4.2 Substitutions

A substitution is represented by a context morphism $[\Gamma \longrightarrow \Delta]$. Note that although a substitution σ can be composed with a dependent type A by using composition of functors, we need to define a fresh notion of composition, noted $A \cdot \sigma$, with the same computational content but with new universe indices to avoid universe inconsistency.

A substitution σ can be extended by an term $a : \text{Tm } (A \cdot \sigma)$ of $A : \text{Typ } \Delta$.

Definition $\text{SubExt } \{ \Gamma \Delta : \text{Context} \} \{ A : \text{Typ } \Delta \} (\sigma : [\Gamma \longrightarrow \Delta]) (a : \text{Tm } (A \cdot \sigma))$
 $: [\Gamma \longrightarrow \Sigma A] := (\lambda \gamma, (\sigma \star \gamma; a \star \gamma) ; \text{SubExt}_{\text{comp}} - -).$

where $\text{SubExt}_{\text{comp}}$ is a proof that it is functorial.

Definition $\text{substF } \{ T \Gamma \} \{ A : \text{Typ } \Gamma \} (F : \text{TypFam } A) (\sigma : [T \longrightarrow \Gamma]) : \text{TypFam } (A \cdot \sigma)$
 $:= ([F \circ \sigma] : \forall t : [T], A \cdot \sigma_{\vdash_s} \star t \longrightarrow \text{Type}_0^1; \text{substF}_{\text{comp}} F \sigma).$

A substitution σ can be applied to a type family F using the composition of a functor with a dependent functor. We abusively note all those different compositions with \circ as it is done in mathematics, whereas they are distinct operators in the COQ development. The weakening substitution of $\Gamma, x : A \vdash$ is given by the first projection.

A type family F in $\text{TypFam } A$ can be partially substituted with an term a in $\text{Tm } A$, noted $F \{ \{a\} \}$, to get its value (a type) at a . This process is defined as $F \{ \{a\} \} := (\lambda \gamma, (F \star \gamma) \star (a \star \gamma) ; -)$ (where $-$ is a proof it is functorial). Note that this pattern of application *up-to a context* γ will be used later to defined other notions of application. Although the computational definitions are the same, the compatibility conditions are always different. This notion of partial substitution in a type family enables to state that A defines a type level λ -abstraction.

Definition $\text{BetaT } \Delta \Gamma (A : \text{Typ } \Gamma) (B : \text{TypDep } A) (\sigma : [\Delta \longrightarrow \Gamma]) (a : \text{Tm } (A \cdot \sigma))$
 $: A B \circ \sigma \{ \{a\} \} \sim_1 B \cdot (\text{SubExt } \sigma a) := (\lambda -, \text{identity } - ; \text{BetaT}_{\text{comp}} - - -).$

4.3 Interpretation of the typing judgment

The typing rules of Figure 1 are interpreted in the groupoid model as described below.

VAR. The rule VAR is given by the second projection plus a proof that the projection is dependently functorial. Note the explicit weakening of A in the returned type. This is because we need to make explicit that the context used to type A is extended with an term of type A . The rule of Figure 1 is more general as it performs an implicit weakening. We do not interpret this part of the rule as weakening is explicit in our model.

Definition $\text{Var } \{ \Gamma \} (A : \text{Typ } \Gamma) : \text{Tm } \uparrow A := (\lambda t, \pi_2 t; \text{Var}_{\text{comp}} A).$

PROD. The rule PROD is interpreted using the dependent functor space, plus a proof that equivalent contexts give rise to isomorphic dependent functor spaces. Note that the rule is defined on type families and not on the dependent type formulation because here we need a fibration point of view.

Definition $\text{Prod } \{I\} (A:\text{Typ } I) (F:\text{TypFam } A)$
 $: \text{Typ } I := (\lambda s, \Pi_0 (F \star s); \text{Prod}_{\text{comp}} A F).$

APP. The rule APP is interpreted using an up-to context application and a proof of dependent functoriality. We abusively note $M \star N$ the application of App .

Definition $\text{App } \{I\} \{A:\text{Typ } I\} \{F:\text{TypFam } A\} (c:\text{Tm } (\text{Prod } F)) (a:\text{Tm } A)$
 $: \text{Tm } (F \{a\}) := (\lambda s, (c \star s) \star (a \star s); \text{App}_{\text{comp}} c a).$

LAM. Term-level λ -abstraction is defined with the same computational meaning as type-level λ -abstraction, but it differs on the proof of dependent functoriality. Note that we use λ in the definition because we need both the fibration (for Prod) and the morphism (for $\text{Tm } B$) point of view.

Definition $\text{Lam } \{I\} \{A:\text{Typ } I\} \{B:\text{TypDep } A\} (b:\text{Tm } B)$
 $: \text{Tm } (\text{Prod } (A B)) := (\lambda \gamma, (\lambda t, b \star (\gamma ; t) ; -); \text{Lam}_{\text{comp}} b).$

SIGMA, PAIR and PROJS. The rules for Σ types are interpreted using the dependent sum Σ on setoids.

Definition $\text{Sigma } \{I\} (A:\text{Typ } I) (F:\text{TypFam } A)$
 $: \text{Typ } I := (\lambda \gamma, [\Gamma], \Sigma (F \star \gamma); \text{Sigma}_{\text{comp}} A F).$

Pairing and projections are obtained by a context lift of pairing and projection of the underlying dependent sum.

CONV. It is not possible to prove in COQ that the conversion rule is preserved because the application of this rule is implicit and can not be reified. Nevertheless, to witness this preservation, we show that beta conversion is valid as a definitional equality on the first projection. As conversion is only done on types and interpretation of types is always projected, this is enough to guarantee that the conversion rule is also admissible.

Definition $\text{Beta } \{I\} \{A:\text{Typ } I\} \{F:\text{TypDep } A\} (b:\text{Tm } F) (a:\text{Tm } A)$
 $: [\text{Lam } b \star a] = [b \circ \text{SubExtId } a] := \text{eq_refl } _.$

where SubExtId is a specialization of SubExt with the identity substitution.

4.4 Identity Types

One of the main interest of the groupoid interpretation is that it allows to interpret a type directed notion of equality which validates the J eliminator of identity types but also various extensional principles. For any terms a and b of a dependent type $A:\text{Typ } I$, we note $\text{ld } a b$ the equality type between a and b obtained by lifting \sim_1 to get a type depending on I .

Definition $\text{ld } \{I\} (A:\text{Typ } I) (a b : \text{Tm } A)$
 $: \text{Typ } I := (\lambda \gamma, (a \star \gamma \sim_1 b \star \gamma ; -); \text{ld}_{\text{comp}} A a b).$

The introduction rule of identity types which corresponds to reflexivity is interpreted by the (lifting of) identity of the underlying setoid.

Definition $\text{Refl } \Gamma (A : \text{Typ } \Gamma) (a : \text{Tm } A)$
 $: \text{Tm } (\text{Id } a \ a) := (\lambda \gamma, \text{identity } (a \star \gamma); \text{Refl}_{\text{comp}} \ - \ -).$

We can interpret the J eliminator of MLTT on Id using functoriality of P and of product (Π_{comp}). In the definition of J, the predicate P depends on the proof of equality, which is interpreted using a Sigma type. The functoriality of P is used on the term $\text{J_Pair } e \ P \ \gamma$, which is a proof that $(a; \text{Refl } a)$ is equal to $(b; e)$. The notation $\uparrow A$ is used to convert the type of terms according to equality on A .

Definition $\text{J } \Gamma (A : \text{Typ } \Gamma) (a \ b : \text{Tm } A) (P : \text{TypFam } (\text{Sigma } (\lambda (Id (a \circ \text{Sub}) (\text{Var } A))))$
 $(e : \text{Tm } (\text{Id } a \ b)) (p : \text{Tm } (P \{ \{ \text{Pair } \uparrow (\text{Refl } a) \} \})))$
 $: \text{Tm } (P \{ \{ \text{Pair } \uparrow e \} \}) := \Pi_{\text{comp}} (\lambda \gamma, (\text{map } (P \star \gamma) (\text{J_Pair } e \ P \ \gamma)); \text{J}_{\text{comp}} \ - \ -) \star p.$

4.5 Functional Extensionality

As the whole interpretation is functorial with respect to a context, the naturality condition required on equality between dependent functors can be deduced from the existence of a transformation. This allows to state dependent functional extensionality.

Definition $\text{FunExt } \Gamma (A : \text{Typ } \Gamma) (F : \text{TypDep } A) (M \ N : \text{Tm } (\text{Prod } (\lambda F)))$
 $(\alpha : \uparrow M \star \text{Var } A \sim_1 \uparrow N \star \text{Var } A) : M \sim_1 N.$
 where $\uparrow M$ is the weakening for terms.

5 Related Work and Conclusion

We have presented an internalization of the groupoid interpretation of Martin-Löf type theory with one universe respecting the invariance under isomorphism principle in CoQ's type theory with universe polymorphism.

The groupoid interpretation is due to Hofmann and Streicher [4]. This interpretation is based on the notion of categories with families introduced by Dybjer [16]. This framework has recently been used by Coquand et al. to give an interpretation in semi-simplicial sets and cubical sets [17,18]. Although very promising, the interpretation based on cubical sets has not yet been mechanically checked and only an extraction procedure based on it has been implemented in Haskell.

Altenkirch et al. have introduced Observational Type Theory (OTT) [1], an intensional type theory where functional extensionality is native, but equality in the universe is structural. To prove expected properties on OTT such as strong normalization, decidable typechecking and canonicity, they use embeddings into Agda and extensional type theory. A setoid interpretation clearly guides their design, and our model could be adapted to interpret this theory as well.

We have strived for generality in our definitions and while our interpretation is done for groupoids, it illustrates the main structures of the model and should adapt to higher dimensional models, specifically ω -groupoids. The next step of our work is to generalize the construction to higher dimensions. We already have a formalization of weak 2-groupoids based on inductive definitions (*à la* enriched categories) on the computational structure. The only mechanism

that is not inductive is the generation of higher-order compatibilities between coherence maps. This is because category enrichment provides a co-inductive definition for *strict* ω -groupoids only. Formalizing in COQ the recent work of Cheng and Leinster [19] on weak enrichment should provide a way to define *weak* ω -groupoids co-inductively using operads to parameterize the compatibility required on coherence maps at higher levels.

References

1. Altenkirch, T., McBride, C., Swierstra, W.: **Observational Equality, Now!** In: PLPV’07, Freiburg, Germany (2007)
2. Voevodsky, V. In: **Univalent Foundations of Mathematics**. Volume 6642. Springer Berlin Heidelberg (2011) 4–4
3. Pelayo, Á., Warren, M.A.: **Homotopy type theory and Voevodsky’s univalent foundations**. (10 2012)
4. Hofmann, M., Streicher, T.: **The Groupoid Interpretation of Type Theory**. In: Twenty-five years of constructive type theory (Venice, 1995). Volume 36 of Oxford Logic Guides. Oxford Univ. Press, New York (1998) 83–111
5. Licata, D.R., Harper, R.: **Canonicity for 2-dimensional type theory**. In Field, J., Hicks, M., eds.: POPL, ACM (2012) 337–348
6. Altenkirch, T., Rypacek, O.: **A Syntactical Approach to Weak omega-Groupoids**. In Cégielski, P., Durand, A., eds.: CSL. Volume 16 of LIPIcs., Schloss Dagstuhl - Leibniz-Zentrum fuer Informatik (2012) 16–30
7. The Univalent Foundations Program: **Homotopy Type Theory: Univalent Foundations for Mathematics**, Institute for Advanced Study (2013)
8. Hoffman, M.: **Syntax and Semantics of Dependent Types**. In: Semantics and Logics of Computation. (1997) 241–298
9. Sozeau, M., Oury, N.: **First-Class Type Classes**. In Otmane Ait Mohamed, C.M., Tahar, S., eds.: TPHOLs. Volume 5170 of LNCS., Springer (August 2008) 278–293
10. Sozeau, M.: **Program-ing Finger Trees in Coq**. In: ICFP’07, Freiburg, Germany, ACM Press (2007) 13–24
11. The Coq development team: **Coq 8.4 Reference Manual**. Inria. (2012)
12. Sozeau, M., Tabareau, N.: **Universe Polymorphism in Coq**. Submitted (January 2014)
13. Garillot, F.: **Generic Proof Tools and Finite Group Theory**. PhD thesis, Ecole Polytechnique X (December 2011)
14. The HoTT Development Team: **Homotopy Type Theory in Coq**
15. Bauer, A., LeFanu Lumsdaine, P.: **A Coq proof that Univalence Axioms implies Functional Extensionality**. (2011)
16. Dybjer, P.: **Internal type theory**. In Berardi, S., Coppo, M., eds.: Types for Proofs and Programs. Volume 1158 of LNCS. Springer Berlin Heidelberg (1996) 120–134
17. Barras, B., Coquand, T., Huber, S.: **A Generalization of Takeuti-Gandy Interpretation**. (2013)
18. Bezem, M., Coquand, T., Huber, S.: **A Model of Type Theory in Cubical Sets**. (December 2013)
19. Cheng, E., Leinster, T.: **Weak ω -categories via terminal coalgebras**. (2012)